
IND8200_ORGANISATION INDUSTRIELLE

TP

Simulation, inventaire et entreposage

Ce document est organisé comme suit :

- A. Cas pratique**
- B. Travail à faire**
- C. Travail à remettre**
- Annexes**

A. Cas pratique

Le propriétaire d'un magasin d'appareils électroniques se pose de sérieuses questions au sujet de son système d'entrepôt et d'approvisionnement. En effet, il arrive régulièrement que les stocks de moniteurs d'un type d'ordinateurs très populaire soient en rupture d'inventaire ("stockouts"). Or, cela est très coûteux pour le magasin : **en plus de perdre le profit de 54 \$ par moniteur**, le magasin perd le client qui ira acheter le moniteur chez un concurrent. Le propriétaire mesure les effets des "stockouts" sur son entreprise en termes **de niveau de service**, c'est-à-dire le pourcentage de demandes satisfaites.

La politique du magasin a toujours été de commander 50 moniteurs à chaque fois que l'inventaire de fin de journée est inférieur à 28 unités. Le propriétaire plaçait alors la commande le lendemain matin. Exemple : si le délai de livraison est d'environ 4 jours. Alors, si l'inventaire à la fin du jour 1 est inférieur à 28, la commande sera placée au début du jour 2 et elle arrivera au début du jour 6.

Le magasin vend en moyenne 6 moniteurs par jour. Par contre, les ventes varient de jour en jour. En révisant ses ventes au cours des derniers mois, le propriétaire a pu déterminer la distribution des probabilités de la demande journalière comme suit :

Demande	0	1	2	3	4	5	6	7	8	9	10
Probabilité	0.01	0.02	0.04	0.06	0.09	0.14	0.18	0.23	0.15	0.06	0.02

De la même façon, il a pu déterminer la distribution de probabilités du délai de livraison du fournisseur qui se situe à une centaine de kilomètres du magasin :

Délai de livraison (jours)	2	4	6
Probabilité	0.3	0.5	0.2

Le propriétaire veut maintenant augmenter son point de commande de façon à obtenir un niveau de service moyen de 98 %.

B. Travail à faire

Pour résoudre ce problème, vous devez simplement suivre les instructions qui suivent et les appliquer en utilisant le fichier intitulé '**GabaritLabSim.xlsx**' disponible sur Moodle.

Pour solutionner ce cas, il faut construire un modèle qui permet d'imiter la dynamique de l'inventaire des moniteurs pour un mois de 30 jours. Ce modèle doit tenir compte de l'aspect aléatoire des demandes journalières et des délais de livraison. Ce problème a pour but de vous familiariser avec la simulation.

1. Générateurs de nombres aléatoire (GNA)

Nous allons créer des générateurs de nombres aléatoire (GNA) pour les demandes et les délais de livraison.

- Exemple : pour simuler les délais de livraison aléatoires, nous devons obtenir un GNA qui va retourner la valeur 2 avec une probabilité de 0.3, la valeur 4 avec une probabilité de 0.5 et la valeur 6 avec une probabilité de 0.2. La fonction ALEA() retourne un nombre aléatoire uniformément distribué entre 0 et 1. En séparant ce nombre selon chaque probabilité, nous obtenons les limites suivantes:

Limite inférieure	Limite supérieure	Délai de livraison
0.0	0.3	2
0.3	0.8	4
0.8	1	6

Chaque intervalle correspond directement à la probabilité de chaque valeur des délais. La même logique est utilisée pour séparer le nombre aléatoire en intervalles de 0 à 1 pour chaque valeur de demande.

- En considérant les données de la feuille "**probabilites**", nous pouvons maintenant utiliser les formules suivantes pour générer des valeurs de demandes et de délais selon les probabilités correspondantes :

GNA (délai de livraison) = RECHERCHEV(ALEA();probabilites!\$B\$5:\$D\$7;3)

GNA (demandes) = RECHERCHEV(ALEA();probabilites!\$G\$5:\$I\$15;3)

- Dans le cas du délai de livraison, le GNA génère un nombre entre 0 et 1 et cherche cette valeur dans la colonne B, la première colonne dans les cellules B5 à D7. Si on ne

peut associer exactement une valeur au nombre généré, RECHERCHEV choisit la ligne de la colonne B qui contient la plus grande valeur qui est plus petite que la valeur du nombre généré par ALEA(). Le dernier argument de la fonction RECHERCHEV (la valeur 3) permet de retourner le résultat dans la troisième colonne (colonne D) de la ligne choisie. Par exemple, si ALEA() génère le nombre 0.63, RECHERCHEV va y associer la valeur 0.3 dans la cellule B6 car c'est la valeur la plus grande qui est inférieure à 0.63. La fonction va retourner la valeur 4 dans la cellule D6.

- Le GNA fonctionne de la même façon pour les demandes journalières.

2. Points de commande

Utilisez la feuille "**parametres**" pour mettre les paramètres de commandes et les futurs résultats de la simulation.

Modèle

Utilisez la feuille de travail "**modele**". La colonne B représente l'inventaire au début de chaque jour, lequel est en fait l'inventaire de la fin de la journée précédente. Les formules de la colonne B sont:

Formule de la cellule B3: =45 (Stock initial)

Formule de la cellule B4: =F3 (à copier dans B5 jusqu'à B32)

- La colonne C représente le nombre d'unités à recevoir pour chaque jour. Nous allons discuter des formules de cette colonne après avoir parlé de celles des colonnes H, I et J, lesquelles sont étroitement reliées entre elles.
- Dans la colonne D, nous utilisons la technique expliquée précédemment pour générer des nombres aléatoires pour les demandes journalières :

*Formule de la cellule D3: =RECHERCHEV(ALEA());probabilites!\$G\$5:\$I\$15;3)
(à copier dans les cellules D4 à D32)*

- Puisqu'il est possible que la demande excède le stock disponible, la colonne E indique combien d'unités de la demande journalière peuvent être vendues. Si la somme de l'inventaire du début (colonne B) et des unités à recevoir (colonne C) est plus grande ou égale à la demande actuelle, alors toute la demande est satisfaite; sinon, le magasin ne peut pas vendre que les unités disponibles:

Formule de la cellule E3: =MIN(B3+C3;D3) (à copier dans les cellules E4 à E32)

- Les valeurs de la colonne F représentent le niveau de l'inventaire à la fin de la journée:
Formule de la cellule F3: =B3+C3-E3 (à copier dans les cellules F4 à F32)

- Pour déterminer si on doit placer ou non une commande, il faut connaître les disponibilités de l'inventaire :

Formule de la cellule G3: =F3

Formule de la cellule G4: =G3-E4+SI(H3=1;parametres!\$F\$7;0)

(à copier dans les cellules G5 à G32)

- La colonne H indique si une commande doit être placée en se basant sur l'inventaire actuel (colonne G) et le point de commande:

Formule de la cellule H3: =SI(G3<parametres!\$F\$5;1;0)

(à copier dans les cellules H4 à H32)

- Si une commande est placée, on doit générer une valeur aléatoire du délai de livraison:

Formule de la cellule I3: =SI(H3=0;0;RECHERCHEV(ALEA();probabilites!\$B\$5:\$D\$7;3))

(à copier dans les cellules I4 à I32)

La formule retourne la valeur 0 si aucune commande n'est placée (si H3=0); sinon, elle donnera une valeur aléatoire (si H3=1).

- Si une commande est placée, la colonne J indique le jour de la réception de la commande en se basant sur la valeur générée du délai.

Formule de la cellule J3: =SI(H3=0;0;A3+1+I3) (à copier dans les cellules I4 à I32)

- Les valeurs de la colonne C sont coordonnées avec celles de la colonne J. Les valeurs non-nulles de la colonne J indiquent les jours au cours desquelles des commandes seront reçues. Par exemple, la cellule **J7** indique qu'une commande va arriver le jour **11**. La valeur 50 de la cellule **C13** représente cette commande et montre qu'elle est bien arrivée:

*=NB.SI(\$J\$3:J12;A13)*parametres!\$F\$7*

Cette formule compte le nombre de fois que la valeur de la cellule A13 (représentant le jour 11) apparaît comme un jour de réception entre les jours 1 à 10 de la colonne J. Cela représente le nombre de commandes à recevoir pendant le jour 11. On multiplie alors ce nombre par la taille de la commande (50), donnée dans la cellule F7 de la feuille "**parametres**". Cela permet de connaître le nombre total d'unités à recevoir au jour 11.

Formule de la cellule C3: =0

*Formule de la cellule C4: =NB.SI(\$J\$3:J3;A4)*parametres!\$F\$7*

(à copier dans les cellules C4 à C32)

- Le niveau de service est calculé dans la cellule E33 en utilisant les valeurs des colonnes D et E comme suit:

Formule de la cellule E33: =SOMME(E3:E32)/SOMME(D3:D32)

3. Réplication du modèle

Le modèle présenté dans la feuille "**modele**" est un scénario possible si le point de commande est de 28 moniteurs d'ordinateurs. Mais nous devons maintenant répliquer le modèle plusieurs fois, et ce, en modifiant les points de commande afin d'augmenter le niveau de service. Revenez à la feuille de travail "**parametres**" pour y répliquer le modèle 200 fois pour chaque point de commande (cellules B11 – B210).

Les cellules C10 à G10 contiennent les valeurs 28, 30, 32, 34 et 36, pour représenter les différents points de commande que le gestionnaire veut analyser en vu d'atteindre le niveau de service désiré (98%). Pour connaître le niveau de service associé à chaque réplication, la formule suivante est entrée dans la cellule B10:

Formule de la cellule B10: = modele!E33

On peut maintenant changer chaque valeur des points de commande et répliquer 200 fois. Cela se fait comme suit:

1. Avec la souris, sélectionnez toutes les cellules de B10 à G210.
2. Cliquez sur le menu Données | Analyse de scénarios | Table de données.
3. Dans la boîte de dialogue de Table, entrez F5 pour la ligne et A1 pour la colonne. Cliquez sur OK.

Excel change chaque valeur des colonnes C10 à G10 dans la cellule F5 du modèle. Pour chaque valeur changée dans la cellule F5, chaque valeur des lignes B11 à B210 sont changées dans la cellule A1 et le tout est recalculé.

Chaque nouvelle entrée que Excel crée en partant par la cellule C11 contient une formule "matricielle" {=TABLE(F5;A1)}. C'est comme cela que la commande Données | Table travaille pour répéter les changements et pour recalculer. Si on ne veut pas attendre à chaque réplication, il faut changer ces formules en valeurs. Voici comment:

1. Sélectionnez les cellules C11 à G210
2. Copiez et collez
3. Cliquez sur Valeurs et OK ensuite
4. Peser sur Enter

Les valeurs de la table sont maintenant des constantes numériques qui ne changeront pas même si la feuille de travail est recalculée via la fonction F9.

C. Travail à remettre

Partie I :

1. Faites une analyse de vos 200 simulations pour vos cinq différents points de commande. À faire
2. Quel est le point de commande idéal si on veut satisfaire notre clientèle à 98%, discuter. À faire
3. Calculer le niveau d'inventaire moyen (calculé à la fin de la journée), pour chacun des cinq différents points de commande? Calculer le coût d'entreposage moyen si le coût de stockage par unité et par jour est de 4\$.
4. Déterminer le nombre de commandes passées. En déduire le coût moyen (journalier) de lancement des commandes, sachant que le lancement d'une commande coûte 20\$.
5. Calculer le coût moyen de pénurie.
6. Calculer le coût total (par jour). Déterminer le point de commande optimale qui minimise le coût total tout en gardant un niveau de service minimum de 98%, discuter.

Pas fait

Partie II :

L'entreprise change de stratégie et décide de fabriquer sur place les produits en question et décide même de ne plus garder de produits en inventaire. Ainsi, les produits seront fabriqués en **juste-à-temps** (le jour même). Si la demande excède la capacité, les unités additionnelles sont produites en temps supplémentaire. L'entreprise se demande s'il vaut mieux embaucher pour avoir une capacité en temps normal de 2, 3, 4, 5, 6, 7, 8, 9, ou 10 unités. Le coût d'une capacité journalière de 1 unité est de 1 u\$, et une unité produite en temps supplémentaire est de 1,5 u\$.

- Modifier le modèle de simulation et faites 200 simulations de chaque combinaison de paramètres, analyser les résultats obtenus.
- Quelle est la meilleure stratégie (partie I ou II), justifier ?

Annexe : fonctions Excel très utiles

RECHERCHEV(valeur_cherchée; table_matrice; no_index_col; valeur_proche)

valeur_cherchée : est la valeur à trouver dans la première colonne de la matrice.

table_matrice : est la table de données dans laquelle est exécutée la recherche.

no_index_col : est le numéro de la colonne d'où la valeur est renvoyée.

valeur_proche : est VRAI par défaut (à garder ainsi)

Table(entrée_en_ligne; entrée_en_colonne)

Cette fonction peut être atteinte en allant dans le menu 'Données/ Table'. Elle s'applique à une table où le coin gauche haut contient une formule et la première ligne et la première colonne contiennent les différentes valeurs d'entrée de notre table. Elle remplace dans la cellule indiquée dans '**entrée_en_ligne**' (respectivement **entrée_en_colonne**) une après l'autre les valeurs de la première ligne de votre table (respectivement les valeurs de la première colonne de votre table) puis à l'intersection de la ligne

Tableau 1 : Exemple d'utilisation de la fonction Table

		<u>variables :</u> x						
		y						
		Différentes valeur de x						
=f(x,y)		1	2	3	4	5	6	7
différentes valeur de y	1	=f(1, 1)	=f(2, 1)	=f(3, 1)	=f(4, 1)	=f(5, 1)	=f(6, 1)	=f(7, 1)
	2	=f(1, 2)	=f(2, 2)	=f(3, 2)	=f(4, 2)	=f(5, 2)	=f(6, 2)	=f(7, 2)
	3	=f(1, 3)	=f(2, 3)	=f(3, 3)	=f(4, 3)	=f(5, 3)	=f(6, 3)	=f(7, 3)
	4	=f(1, 4)	=f(2, 4)	=f(3, 4)	=f(4, 4)	=f(5, 4)	=f(6, 4)	=f(7, 4)
	5	=f(1, 5)	=f(2, 5)	=f(3, 5)	=f(4, 5)	=f(5, 5)	=f(6, 5)	=f(7, 5)
	6	=f(1, 6)	=f(2, 6)	=f(3, 6)	=f(4, 6)	=f(5, 6)	=f(6, 6)	=f(7, 6)
	7	=f(1, 7)	=f(2, 7)	=f(3, 7)	=f(4, 7)	=f(5, 7)	=f(6, 7)	=f(7, 7)
	8	=f(1, 8)	=f(2, 8)	=f(3, 8)	=f(4, 8)	=f(5, 8)	=f(6, 8)	=f(7, 8)
	9	=f(1, 9)	=f(2, 9)	=f(3, 9)	=f(4, 9)	=f(5, 9)	=f(6, 9)	=f(7, 9)
	10	=f(1, 10)	=f(2, 10)	=f(3, 10)	=f(4, 10)	=f(5, 10)	=f(6, 10)	=f(7, 10)
	11	=f(1, 11)	=f(2, 11)	=f(3, 11)	=f(4, 11)	=f(5, 11)	=f(6, 11)	=f(7, 11)
	12	=f(1, 12)	=f(2, 12)	=f(3, 12)	=f(4, 12)	=f(5, 12)	=f(6, 12)	=f(7, 12)
	13	=f(1, 13)	=f(2, 13)	=f(3, 13)	=f(4, 13)	=f(5, 13)	=f(6, 13)	=f(7, 13)